

0.1 Header

DiracSimplify[exp] simplifies products of Dirac matrices in **exp** and expands noncommutative products. The simplifications are done by applying **Contract**, **DiracEquation**, **DiracTrick**, **SpinorChainTrick** and **SirlinSimplify**. All γ^5 , γ^6 and γ^7 are moved to the right. The order of the other Dirac matrices is not changed, unless the option **DiracOrder** is set to **True**.

0.1.1 See also

[Overview](#), [Contract](#), [DiracEquation](#), [DiracSigmaExplicit](#), [DiracSubstitute5](#), [DiracSubstitute67](#), [DiracGamma](#), [DiracGammaExpand](#), [DiracOrder](#), [DiracTrace](#), [DiracTraceEvaluate](#), [DiracTrick](#), [FCDiracIsolate](#), [SirlinSimplify](#), [SpinorChainTrick](#), [SpinorChainEvaluate](#), [ToDiracGamma67](#).

0.1.2 Examples

Simplify a 4-dimensional Dirac matrix chain with a dummy Lorentz index

```
GA[\[Mu], \[Nu], \[Mu]]
DiracSimplify[%]
```

$$\bar{\gamma}^{\mu} \cdot \bar{\gamma}^{\nu} \cdot \bar{\gamma}^{\mu}$$

$$-2\bar{\gamma}^{\nu}$$

Another common simplification concerns Dirac matrices contracted to the same 4-vector

```
GS[p] . GS[p]
DiracSimplify[%]
```

$$(\bar{\gamma} \cdot \bar{p}) \cdot (\bar{\gamma} \cdot \bar{p})$$

$$\bar{p}^2$$

Unlike **DiracTrick**, **DiracSimplify** also carries out noncommutative expansions

```
GS[a + b] . GS[p] . GS[c + d] . GS[p]
DiracSimplify[%]
```

$$(\bar{\gamma} \cdot (\bar{a} + \bar{b})) \cdot (\bar{\gamma} \cdot \bar{p}) \cdot (\bar{\gamma} \cdot (\bar{c} + \bar{d})) \cdot (\bar{\gamma} \cdot \bar{p})$$

$$2(\bar{c} \cdot \bar{p})(\bar{\gamma} \cdot \bar{a}) \cdot (\bar{\gamma} \cdot \bar{p}) - \bar{p}^2(\bar{\gamma} \cdot \bar{a}) \cdot (\bar{\gamma} \cdot \bar{c}) + 2(\bar{d} \cdot \bar{p})(\bar{\gamma} \cdot \bar{a}) \cdot (\bar{\gamma} \cdot \bar{p}) - \bar{p}^2(\bar{\gamma} \cdot \bar{a}) \cdot (\bar{\gamma} \cdot \bar{d}) \\ + 2(\bar{c} \cdot \bar{p})(\bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{p}) - \bar{p}^2(\bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{c}) + 2(\bar{d} \cdot \bar{p})(\bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{p}) - \bar{p}^2(\bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{d})$$

```
DiracTrick[GS[a + b] . GS[p] . GS[c + d] . GS[p]]
```

$$2(\bar{\gamma} \cdot (\bar{a} + \bar{b})) \cdot (\bar{\gamma} \cdot \bar{p})((\bar{c} + \bar{d}) \cdot \bar{p}) - \bar{p}^2(\bar{\gamma} \cdot (\bar{a} + \bar{b})) \cdot (\bar{\gamma} \cdot (\bar{c} + \bar{d}))$$

Some of those expansions can be inhibited via the option **Expanding**.

```
DiracSimplify[GS[a + b] . GS[p] . GS[c + d] . GS[p], Expanding -> False]
```

$$-\bar{p}^2(\bar{\gamma} \cdot \bar{a} + \bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{c} + \bar{\gamma} \cdot \bar{d}) + 2(\bar{c} \cdot \bar{p})(\bar{\gamma} \cdot \bar{a} + \bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{p}) + 2(\bar{d} \cdot \bar{p})(\bar{\gamma} \cdot \bar{a} + \bar{\gamma} \cdot \bar{b}) \cdot (\bar{\gamma} \cdot \bar{p})$$

The matrix chain may also live in D dimensions

```
GAD[\[Mu], \[Nu], \[Mu]]
```

```
DiracSimplify[%]
```

$$\gamma^\mu \cdot \gamma^\nu \cdot \gamma^\mu$$

$$2\gamma^\nu - D\gamma^\nu$$

```
GSD[p] . GAD[\[Alpha], \[Beta]] . GSD[p]
```

```
DiracSimplify[%]
```

$$(\gamma \cdot p) \cdot \gamma^\alpha \cdot \gamma^\beta \cdot (\gamma \cdot p)$$

$$p^2 \gamma^\alpha \cdot \gamma^\beta + 2p^\alpha \gamma^\beta \cdot (\gamma \cdot p) - 2p^\beta \gamma^\alpha \cdot (\gamma \cdot p)$$

```
GAD @@ Join[{\[Mu]}, Table[Subscript[\[Nu], i], {i, 6}], {\[Mu]}]
```

```
DiracSimplify[%]
```

$$\gamma^\mu \cdot \gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} \cdot \gamma^\mu$$

$$\begin{aligned} & -12\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} + D\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} + 4\gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_1 \nu_2} \\ & - 4\gamma^{\nu_2} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_1 \nu_3} + 4\gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_1 \nu_4} - 4\gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_6} g^{\nu_1 \nu_5} \\ & + 4\gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} g^{\nu_1 \nu_6} + 4\gamma^{\nu_1} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_2 \nu_3} - 4\gamma^{\nu_1} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_2 \nu_4} + 4\gamma^{\nu_1} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_6} g^{\nu_2 \nu_5} \\ & - 4\gamma^{\nu_1} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} g^{\nu_2 \nu_6} + 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_5} \cdot \gamma^{\nu_6} g^{\nu_3 \nu_4} - 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_6} g^{\nu_3 \nu_5} + 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_4} \cdot \gamma^{\nu_5} g^{\nu_3 \nu_6} \\ & + 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_6} g^{\nu_4 \nu_5} - 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_5} g^{\nu_4 \nu_6} + 4\gamma^{\nu_1} \cdot \gamma^{\nu_2} \cdot \gamma^{\nu_3} \cdot \gamma^{\nu_4} g^{\nu_5 \nu_6} \end{aligned}$$

```
-1/2 GA[5] . (GAD[\[Mu]] . GSD[v] - FVD[v, \[Mu]]) FVD[v, \[Mu]]
```

```
DiracSimplify[%]
```

$$-\frac{1}{2}v^\mu \bar{\gamma}^5 \cdot (\gamma^\mu \cdot (\gamma \cdot v) - v^\mu)$$

$$0$$

γ^5 and the chirality projectors are always moved to the right

```
GA[5, \[Mu], \[Nu]]
```

```
DiracSimplify[%]
```

$$\bar{\gamma}^5 \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\nu$$

$$\bar{\gamma}^\mu \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^5$$

```
GA[6] . GS[p + q]
```

```
DiracSimplify[%]
```

$$\bar{\gamma}^6 \cdot (\bar{\gamma} \cdot (\bar{p} + \bar{q}))$$

$$(\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^7 + (\bar{\gamma} \cdot \bar{q}) \cdot \bar{\gamma}^7$$

The properties of the chirality projectors are taken into account without substituting explicit expressions for γ^6 and γ^7 .

```
GA[\[Mu]] . (c1 GA[6] + c2 GA[7]) . (GA[p] + m) . (c3 GA[6] + c4 GA[7]) .
GA[\[Mu]]
DiracSimplify[%]
```

$$\bar{\gamma}^{\mu} \cdot (c1 \bar{\gamma}^6 + c2 \bar{\gamma}^7) \cdot (\bar{\gamma}^p + m) \cdot (c3 \bar{\gamma}^6 + c4 \bar{\gamma}^7) \cdot \bar{\gamma}^{\mu}$$

$$4 c1 c3 m \bar{\gamma}^7 - 2 c1 c4 \bar{\gamma}^p \cdot \bar{\gamma}^6 - 2 c2 c3 \bar{\gamma}^p \cdot \bar{\gamma}^7 + 4 c2 c4 m \bar{\gamma}^6$$

Moreover, $\frac{1}{2} (1 \pm \gamma^5)$ is automatically replaced by $\gamma^{6/7}$.

```
(1/2 - GA[5]/2) . (-((a + GS[p + q])/b)) . (1/2 + GA[5]/2)
DiracSimplify[%]
```

$$\left(\frac{1}{2} - \frac{\bar{\gamma}^5}{2}\right) \cdot \left(-\frac{\bar{\gamma} \cdot (\bar{p} + \bar{q}) + a}{b}\right) \cdot \left(\frac{\bar{\gamma}^5}{2} + \frac{1}{2}\right)$$

$$-\frac{(\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^6}{b} - \frac{(\bar{\gamma} \cdot \bar{q}) \cdot \bar{\gamma}^6}{b}$$

Suitable combinations of γ^5 will not be rewritten in terms of chirality projectors, if the option **ToDiracGamma67** is set to **False**.

```
DiracSimplify[(1/2 - GA[5]/2) . (-((a + GS[p + q])/b)) . (1/2 + GA[5]/2),
  ToDiracGamma67 -> False]
```

$$-\frac{\bar{\gamma} \cdot \bar{p}}{2b} - \frac{(\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^5}{2b} - \frac{\bar{\gamma} \cdot \bar{q}}{2b} - \frac{(\bar{\gamma} \cdot \bar{q}) \cdot \bar{\gamma}^5}{2b}$$

However, if the final result must contain only γ^5 but not γ^6 or γ^7 , it is better to invoke the option **DiracSubstitute67**. This way DiracSimplify can perform more intermediate simplifications before presenting the final result.

```
DiracSimplify[(1/2 - GA[5]/2) . (-((a + GS[p + q])/b)) . (1/2 + GA[5]/2),
  DiracSubstitute67 -> True]
```

$$-\frac{\bar{\gamma} \cdot \bar{p}}{2b} - \frac{(\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^5}{2b} - \frac{\bar{\gamma} \cdot \bar{q}}{2b} - \frac{(\bar{\gamma} \cdot \bar{q}) \cdot \bar{\gamma}^5}{2b}$$

It is also possible to eliminate γ^5 by rewriting it in terms of the chirality projectors

```
DiracSimplify[GA[5, \[Mu], \[Nu]], DiracSubstitute5 -> True]
```

$$\bar{\gamma}^{\mu}.\bar{\gamma}^{\nu}.\bar{\gamma}^6 - \bar{\gamma}^{\mu}.\bar{\gamma}^{\nu}.\bar{\gamma}^7$$

The Dirac equation is routinely used to simplify closed spinor chains.

```
(SpinorVBar[Subscript[p, 2], Subscript[m, 2]] . (GS[Subscript[p, 1]] +
  Subscript[m, 1]) . SpinorU[Subscript[p, 1], Subscript[m, 1]])
DiracSimplify[%]
```

$$\bar{v}(p_2, m_2) . (\bar{\gamma} . \bar{p}_1 + m_1) . u(p_1, m_1)$$

$$2m_1 (\varphi(-\bar{p}_2, m_2)) . (\varphi(\bar{p}_1, m_1))$$

```
SpinorVBar[p] . GS[p] . SpinorU[q]
DiracSimplify[%]
```

$$\bar{v}(p) . (\bar{\gamma} . \bar{p}) . u(q)$$

$$0$$

Use the option **DiracEquation** to deactivate this type of simplifications.

```
DiracSimplify[SpinorVBar[p] . GS[p] . SpinorU[q], DiracEquation -> False]
```

$$(\varphi(-\bar{p})) . (\bar{\gamma} . \bar{p}) . (\varphi(\bar{q}))$$

Suitable products of 4-dimensional spinor chains are simplified using Sirlin's identities

```
(SpinorUBar[Subscript[p, 3], Subscript[m, 3]] . GA[\[Mu], \[Rho], \[Nu], 6]
. SpinorU[Subscript[p, 1],
  Subscript[m, 1]] SpinorUBar[Subscript[p, 4],
  Subscript[m, 4]] . GA[\[Mu], \[Tau], \[Nu], 6] . SpinorU[Subscript[p,
  2], Subscript[m, 2]])
DiracSimplify[%]
```

$$\bar{u}(p_3, m_3) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_1, m_1) \bar{u}(p_4, m_4) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_2, m_2)$$

$$(\varphi(\bar{p}_3, m_3)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_2, m_2))$$

The applications of Sirlin's identities can be disabled by setting the option **SirlinSimplify** to **False**.

```
DiracSimplify[SpinorUBar[Subscript[p, 3], Subscript[m, 3]] . GA[\[Mu],
\[Rho], \[Nu],
6] . SpinorU[Subscript[p, 1], Subscript[m, 1]]*
SpinorUBar[Subscript[p, 4], Subscript[m, 4]] . GA[\[Mu], \[Tau], \[Nu],
6] . SpinorU[Subscript[p, 2], Subscript[m, 2]], SirlinSimplify ->
False]
```

$$(\varphi(\bar{p}_3, m_3)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_2, m_2))$$

Even when the usage of Sirlin's identities is disabled, DiracSimplify will still try to perform some simplifications on the spinor chains, e.g. by canonicalizing the dummy indices.

```
(c1 SpinorUBar[Subscript[p, 3], Subscript[m, 3]] . GA[\[Mu], \[Rho], \[Nu],
6] . SpinorU[Subscript[p,
1], Subscript[m, 1]] SpinorUBar[Subscript[p, 4], Subscript[m,
4]] . GA[\[Mu], \[Tau], \[Nu], 6] . SpinorU[Subscript[p, 2],
Subscript[m, 2]] +
c2 SpinorUBar[Subscript[p, 3], Subscript[m, 3]] . GA[\[Alpha], \[Rho],
\[Nu], 6] . SpinorU[Subscript[p, 1], Subscript[m, 1]]
SpinorUBar[Subscript[p,
4], Subscript[m, 4]] . GA[\[Alpha], \[Tau], \[Nu], 6] .
SpinorU[Subscript[p, 2], Subscript[m, 2]])

DiracSimplify[%, SirlinSimplify -> False] // Factor
```

$$c1 \bar{u}(p_3, m_3) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_1, m_1) \bar{u}(p_4, m_4) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_2, m_2) + c2 \bar{u}(p_3, m_3) \cdot \bar{\gamma}^\alpha \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_1, m_1) \bar{u}(p_4, m_4) \cdot \bar{\gamma}^\alpha \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot u(p_2, m_2)$$

$$c1 (\varphi(\bar{p}_3, m_3)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) \cdot \bar{\gamma}^\mu \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_2, m_2)) + c2 (\varphi(\bar{p}_3, m_3)) \cdot \bar{\gamma}^\alpha \cdot \bar{\gamma}^\rho \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) \cdot \bar{\gamma}^\alpha \cdot \bar{\gamma}^\tau \cdot \bar{\gamma}^\nu \cdot \bar{\gamma}^6 \cdot (\varphi(\bar{p}_2, m_2))$$

Setting **SpinorChainTrick** to 'False' disables this behavior.

```
DiracSimplify[c1 SpinorUBar[Subscript[p, 3], Subscript[m, 3]] . GA[\[Mu],
\[Rho],
  \[Nu], 6] . SpinorU[Subscript[p, 1], Subscript[m, 1]]
  SpinorUBar[Subscript[p,
    4], Subscript[m, 4]] . GA[\[Mu], \[Tau], \[Nu], 6] .
  SpinorU[Subscript[p, 2],
    Subscript[m, 2]] + c2 SpinorUBar[Subscript[p, 3], Subscript[m,
    3]] . GA[\[Alpha], \[Rho], \[Nu], 6] . SpinorU[Subscript[p, 1],
    Subscript[m,
    1]] SpinorUBar[Subscript[p, 4], Subscript[m, 4]] . GA[\[Alpha],
    \[Tau],
    \[Nu], 6] . SpinorU[Subscript[p, 2], Subscript[m, 2]],
SirlinSimplify -> False, SpinorChainTrick -> False]
```

$$c1 (\varphi(\bar{p}_3, m_3)) . \bar{\gamma}^\mu . \bar{\gamma}^\rho . \bar{\gamma}^\nu . \bar{\gamma}^6 . (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) . \bar{\gamma}^\mu . \bar{\gamma}^\tau . \bar{\gamma}^\nu . \bar{\gamma}^6 . (\varphi(\bar{p}_2, m_2)) + c2 (\varphi(\bar{p}_3, m_3)) . \bar{\gamma}^\alpha . \bar{\gamma}^\rho . \bar{\gamma}^\nu . \bar{\gamma}^6 . (\varphi(\bar{p}_1, m_1)) (\varphi(\bar{p}_4, m_4)) . \bar{\gamma}^\alpha . \bar{\gamma}^\tau . \bar{\gamma}^\nu . \bar{\gamma}^6 . (\varphi(\bar{p}_2, m_2))$$

DiracSimplify will not reorder Dirac matrices lexicographically, but can be forced to do so via the option **DiracOrder**.

```
DiracSimplify[GA[\[Nu], \[Mu]]]
DiracSimplify[GA[\[Nu], \[Mu]], DiracOrder -> True]
```

$$\bar{\gamma}^\nu . \bar{\gamma}^\mu$$

$$2\bar{g}^{\mu\nu} - \bar{\gamma}^\mu . \bar{\gamma}^\nu$$

Setting **InsideDiracTrace** to **True** makes the function assume that it is acting inside a Dirac trace. For instance, chains with an odd number of Dirac matrices will be set to zero.

```
GA[\[Mu], \[Nu], \[Rho]]
DiracSimplify[%, InsideDiracTrace -> True]
```

$$\bar{\gamma}^\mu . \bar{\gamma}^\nu . \bar{\gamma}^\rho$$

$$0$$

Yet, it will not explicitly calculate the trace

```
GA[\[Mu], \[Nu], \[Rho], \[Sigma]]
DiracSimplify[%, InsideDiracTrace -> True]
```

$$\bar{\gamma}^{\mu}.\bar{\gamma}^{\nu}.\bar{\gamma}^{\rho}.\bar{\gamma}^{\sigma}$$

$$\bar{\gamma}^{\mu}.\bar{\gamma}^{\nu}.\bar{\gamma}^{\rho}.\bar{\gamma}^{\sigma}$$

Since FeynCalc 9.3, **DiracSimplify** will automatically evaluate Dirac traces in the input expression

```
DiracTrace[GA[\[Mu], \[Nu], \[Rho], \[Sigma]]]
DiracSimplify[%]
```

$$\text{tr}(\bar{\gamma}^{\mu}.\bar{\gamma}^{\nu}.\bar{\gamma}^{\rho}.\bar{\gamma}^{\sigma})$$

$$4\bar{g}^{\mu\sigma}\bar{g}^{\nu\rho} - 4\bar{g}^{\mu\rho}\bar{g}^{\nu\sigma} + 4\bar{g}^{\mu\nu}\bar{g}^{\rho\sigma}$$

```
DiracTrace[(-GSD[q] + SMP["m_e"]) . GAD[\[Nu]] . (GSD[p - q] + SMP["m_e"])
. GAD[\[Mu]]]
DiracSimplify[%]
```

$$\text{tr}((m_e - \gamma \cdot q) \cdot \gamma^{\nu} \cdot (m_e + \gamma \cdot (p - q)) \cdot \gamma^{\mu})$$

$$4m_e^2 g^{\mu\nu} + 4g^{\mu\nu}(p \cdot q) - 4q^2 g^{\mu\nu} - 4p^{\nu} q^{\mu} - 4p^{\mu} q^{\nu} + 8q^{\mu} q^{\nu}$$

This will not happen if the option **DiracTraceEvaluate** is set to **False**. However, **DiracSimplify** will still perform some simplifications inside the trace, without evaluating it explicitly.

```
DiracSimplify[DiracTrace[(-GSD[q] + SMP["m_e"]) . GAD[\[Nu]] . (GSD[p - q]
+
SMP["m_e"]) . GAD[\[Mu]]] , DiracTraceEvaluate -> False]
```

$$\begin{aligned} &\text{tr}(m_e^2 \gamma^{\nu} \cdot \gamma^{\mu} + m_e \gamma^{\nu} \cdot (\gamma \cdot p) \cdot \gamma^{\mu} - m_e \gamma^{\nu} \cdot (\gamma \cdot q) \cdot \gamma^{\mu} - m_e (\gamma \cdot q) \cdot \gamma^{\nu} \cdot \gamma^{\mu} \\ &- (\gamma \cdot q) \cdot \gamma^{\nu} \cdot (\gamma \cdot p) \cdot \gamma^{\mu} - q^2 \gamma^{\nu} \cdot \gamma^{\mu} + 2q^{\nu} (\gamma \cdot q) \cdot \gamma^{\mu}) \end{aligned}$$

Set **DiracTrace** to **False** if you want **DiracSimplify** not to touch the Dirac traces.


```
DiracSimplify[DiracTrace[(-GSD[q] + SMP["m_e"]) . GAD[\[Nu]] . (GSD[p - q]
+
    SMP["m_e"]) . GAD[\[Mu]]] , DiracTraceEvaluate -> False, DiracTrace
-> False]
```

$$\text{tr}((m_e - \gamma \cdot q) \cdot \gamma^\nu \cdot (m_e + \gamma \cdot (p - q)) \cdot \gamma^\mu)$$

When doing calculations at one loop and above, you may encounter expressions that contain D - and 4-dimensional objects.

Although **DiracSimplify** can handle such terms effortlessly, it will not do so unless you certify that you fully understand what you are doing: being sloppy with the dimensions easily leads to inconsistencies and wrong results!

```
GAD[\[Mu]] . (GA[p] + m) . GAD[\[Mu]]
DiracSimplify[%]
```

$$\gamma^\mu \cdot (\bar{\gamma}^P + m) \cdot \gamma^\mu$$

DiracTrace: Expressions that mix D-, 4- and D-4- dimensional quantities are forbidden in Dirac matrix chains unless you are using the t'Hooft-Veltman scheme. For every other scheme, please recheck your input expressions and ensure that all matrices, spinors and tensors are purely D- dimensional. You might want to use FCGetDimensions[exp] to find the offending terms and fix them by hand or employ ChangeDimension[exp,D] to convert the whole expression to D- dimensions. If you explicitly intend to use the t'Hooft- Veltman scheme, please activate it via FCSetDiracGammaScheme["BMHV"].

\$Aborted

By default, FeynCalc uses the naive dimensional regularization (NDR) scheme, where all Dirac matrices are taken to be D -dimensional. Therefore, in NDR you may not have mixtures of Dirac matrices living in D and 4 dimensions. However, such expressions are possible in the t'Hooft-Veltman-Breitenlohner-Maison (BMHV) scheme.

```
FCSetDiracGammaScheme["BMHV"];
DiracSimplify[GAD[\[Mu]] . (GA[p] + m) . GAD[\[Mu]]]
```

$$-D\bar{\gamma}^P + 2\bar{\gamma}^P + Dm$$

```
FCSetDiracGammaScheme["NDR"];
```

The BMHV scheme is a special prescription for handling γ^5 in dimensional regularization. Do not activate this scheme mindlessly just to get rid of errors from DiracSimplify! If you are doing a calculation in NDR or a calculation that does not involve γ^5 , better make sure that your input expressions are correctly written to be D -dimensional objects.

Traces that contain an odd number of γ^5 or chirality projectors cannot be calculated unambiguously in NDR. To avoid inconsistencies, DiracTrace will refuse to evaluate such traces in NDR.

```
DiracTrace[GAD[\[Mu],\[Nu],\[Rho],\[Sigma],\[Alpha],\[Beta]] . GA[5]]
DiracSimplify[%]
```

$$\text{tr}(\gamma^\mu \cdot \gamma^\nu \cdot \gamma^\rho \cdot \gamma^\sigma \cdot \gamma^\alpha \cdot \gamma^\beta \cdot \bar{\gamma}^5)$$

$$\text{tr}(\gamma^\mu \cdot \gamma^\nu \cdot \gamma^\rho \cdot \gamma^\sigma \cdot \gamma^\alpha \cdot \gamma^\beta \cdot \bar{\gamma}^5)$$

Such traces can be consistently calculated in the BMHV scheme. Our scheme choice as of course also possible, but those are not implemented in FeynCalc.

```
FCSetDiracGammaScheme["BMHV"];
DiracSimplify[DiracTrace[GAD[\[Mu],\[Nu],\[Rho],\[Sigma],\[Alpha],
\[Beta]] . GA[5]]]
```

$$\begin{aligned} & -4ig^{\alpha\beta}\bar{\epsilon}^{\mu\nu\rho\sigma} - 4ig^{\alpha\mu}\bar{\epsilon}^{\beta\nu\rho\sigma} + 4ig^{\alpha\nu}\bar{\epsilon}^{\beta\mu\rho\sigma} - 4ig^{\alpha\rho}\bar{\epsilon}^{\beta\mu\nu\sigma} + 4ig^{\alpha\sigma}\bar{\epsilon}^{\beta\mu\nu\rho} \\ & + 4ig^{\beta\mu}\bar{\epsilon}^{\alpha\nu\rho\sigma} - 4ig^{\beta\nu}\bar{\epsilon}^{\alpha\mu\rho\sigma} + 4ig^{\beta\rho}\bar{\epsilon}^{\alpha\mu\nu\sigma} - 4ig^{\beta\sigma}\bar{\epsilon}^{\alpha\mu\nu\rho} - 4ig^{\mu\nu}\bar{\epsilon}^{\alpha\beta\rho\sigma} \\ & + 4ig^{\mu\rho}\bar{\epsilon}^{\alpha\beta\nu\sigma} - 4ig^{\mu\sigma}\bar{\epsilon}^{\alpha\beta\nu\rho} - 4ig^{\nu\rho}\bar{\epsilon}^{\alpha\beta\mu\sigma} + 4ig^{\nu\sigma}\bar{\epsilon}^{\alpha\beta\mu\rho} - 4ig^{\rho\sigma}\bar{\epsilon}^{\alpha\beta\mu\nu} \end{aligned}$$

```
FCSetDiracGammaScheme["NDR"];
```

Keep in mind that the BMHV scheme violates axial Ward identities and requires special model-dependent counter-terms to restore those. Therefore, just setting **FCSetDiracGammaScheme["BMHV"]** does not magically resolve all your troubles with γ^5 in D -dimensions. The proper treatment of γ^5 in dimensional regularization is an intricate issue that cannot be boiled down to a simple and universal recipe. FeynCalc merely carries out the algebraic operations that you request, but it is still your task to ensure that what you do makes sense.

Since FeynCalc 9.3 it is also possible to simplify Dirac matrices with Cartesian or temporal indices. However, the support of nonrelativistic calculations is a very new feature, so that things may not work as smooth as they do for manifestly Lorentz covariant expressions.

```
CGA[i] . CGA[i]
DiracSimplify[%]
```

$$\bar{\gamma}^i \cdot \bar{\gamma}^i$$

$$-3$$

```
CGA[i] . CGS[p] . CGA[j] . CGS[p + q]
DiracSimplify[%]
```

$$\bar{\gamma}^i \cdot (\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^j \cdot (\bar{\gamma} \cdot (\bar{p} + \bar{q}))$$

$$\bar{p}^2 \bar{\gamma}^i \cdot \bar{\gamma}^j - 2 \bar{p}^j \bar{\gamma}^i \cdot (\bar{\gamma} \cdot \bar{p}) + \bar{\gamma}^i \cdot (\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^j \cdot (\bar{\gamma} \cdot \bar{q})$$

```
CGA[i] . CGS[p] . CGA[j] . CGS[p + q] KD[i, j]
DiracSimplify[%]
```

$$\bar{\delta}^{ij} \bar{\gamma}^i \cdot (\bar{\gamma} \cdot \bar{p}) \cdot \bar{\gamma}^j \cdot (\bar{\gamma} \cdot (\bar{p} + \bar{q}))$$

$$(\bar{\gamma} \cdot \bar{p}) \cdot (\bar{\gamma} \cdot \bar{q}) - \bar{p}^2$$

```
TGA[] . CGA[i] . TGA[]
DiracSimplify[%]
```

$$\bar{\gamma}^0 \cdot \bar{\gamma}^i \cdot \bar{\gamma}^0$$

$$-\bar{\gamma}^i$$

```
DiracTrace[CGAD[i, j, k, l]]
DiracSimplify[%]
```

$$\text{tr}(\gamma^i \gamma^j \gamma^k \gamma^l)$$

$$4\delta^{il}\delta^{jk} - 4\delta^{ik}\delta^{jl} + 4\delta^{ij}\delta^{kl}$$

For performance reasons **DiracSimplify** will not canonically order Dirac matrices and canonicalize Lorentz/Cartesian indices by default. However, amplitudes involving 4-fermion operators may require such additional simplifications. In this case they should explicitly activated by the user. Of course, this will somewhat slow down the evaluation.

```
ex = (Spinor[-Momentum[p1, D], mb, 1] . GAD[\[Mu]] . GA[7] . GAD[\[Nu]] .
GAD[\[Alpha]] .
  GAD[\[Beta]] . GAD[\[Delta]] . GA[7] . Spinor[-Momentum[p4, D], 0, 1]
Spinor[Momentum[p3, D], 0,
  1] . GAD[\[Alpha]] . GAD[\[Beta]] . GAD[\[Delta]] . GA[7] .
  GAD[\[Nu]] . GAD[\[Mu]] .
  GA[7] . Spinor[Momentum[p2, D], 0, 1])
DiracSimplify[ex]
```

$$(\varphi(p3)).\gamma^\alpha.\gamma^\beta.\gamma^\delta.\bar{\gamma}^7.\gamma^\nu.\gamma^\mu.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1,mb)).\gamma^\mu.\bar{\gamma}^7.\gamma^\nu.\gamma^\alpha.\gamma^\beta.\gamma^\delta.\bar{\gamma}^7.(\varphi(-p4))$$

$$(\varphi(p3)).\gamma^\alpha.\gamma^\beta.\gamma^\delta.\gamma^\nu.\gamma^\mu.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1,mb)).\gamma^\mu.\gamma^\nu.\gamma^\alpha.\gamma^\beta.\gamma^\delta.\bar{\gamma}^7.(\varphi(-p4))$$

```
DiracSimplify[ex, DiracOrder -> True, LorentzIndexNames -> {i1, i2, i3, i4,
i5}]
```

$$\begin{aligned} & -24D^2(\varphi(p3)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1, \\ & mb)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(-p4)) + 14D(\varphi(p3)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1, \\ & mb)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\bar{\gamma}^7.(\varphi(-p4)) + 112D(\varphi(p3)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1, \\ & mb)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(-p4)) - (\varphi(p3)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\gamma^{i4}.\gamma^{i5}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1, \\ & mb)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\gamma^{i4}.\gamma^{i5}.\bar{\gamma}^7.(\varphi(-p4)) - 36(\varphi(p3)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1, \\ & mb)).\gamma^{i1}.\gamma^{i2}.\gamma^{i3}.\bar{\gamma}^7.(\varphi(-p4)) - 64(\varphi(p3)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(p2))(\varphi(-p1,mb)).\gamma^{i1}.\bar{\gamma}^7.(\varphi(-p4)) \end{aligned}$$

DiracSimplify automatically evaluates suitable spinor products with equal momenta, e.g.

```
ex = SpinorUBar[p, m] . SpinorU[p, m]
DiracSimplify[ex]
```

$$\bar{u}(p, m).u(p, m)$$

$$2m$$

This behavior can be turned off by setting the option **SpinorChainEvaluate** to **False**

```
| DiracSimplify[ex, SpinorChainEvaluate -> False]
```

$$(\varphi(\bar{p}, m)) . (\varphi(\bar{p}, m))$$